<commit_before>{ stdenv, fetchurl }:

stdenv.mkDerivation rec {
  name = "pm-utils-1.3.0";

  src = fetchurl {
    url = "http://pm-utils.freedesktop.org/releases/${name}.tar.gz";
    md5 = "37d71f8adbb409442212a85a080d324d";
  };

  configureFlags = "--sysconfdir=/etc";

  preConfigure =
    ''
      # Install the manpages (xmlto isn't really needed).
      substituteInPlace man/Makefile.in --replace '@HAVE_XMLTO_TRUE@' ""

      # Don't screw up the PATH.
      substituteInPlace pm/pm-functions.in --replace '/sbin:/usr/sbin:/bin:/usr/bin' '$PATH'

      substituteInPlace pm/sleep.d/00logging --replace /bin/uname "$(type -P uname)"
    '';

  meta = {
    homepage = http://pm-utils.freedesktop.org/wiki/;
    description = "A small collection of scripts that handle suspend and resume on behalf of HAL";
    license = "GPLv2";
  };
}<commit_msg>* pm-utils updated to 1.4.1.  Also put all runtime dependencies in its   $PATH (because upower is called by dbus which doesn't allow setting   the $PATH).<commit_after>{ stdenv, fetchurl, coreutils, gnugrep, utillinux, module_init_tools
, procps, kbd }:

let

  binPath = stdenv.lib.makeSearchPath "bin"
    [ coreutils gnugrep utillinux module_init_tools procps kbd ];

  sbinPath = stdenv.lib.makeSearchPath "sbin"
    [ procps ];
    
in 

stdenv.mkDerivation rec {
  name = "pm-utils-1.4.1";

  src = fetchurl {
    url = "http://pm-utils.freedesktop.org/releases/${name}.tar.gz";
    sha256 = "02qc6zaf7ams6qcc470fwb6jvr4abv3lrlx16clqpn36501rkn4f";
  };

  configureFlags = "--sysconfdir=/etc";

  preConfigure =
    ''
      # Install the manpages (xmlto isn't really needed).
      substituteInPlace man/Makefile.in --replace '@HAVE_XMLTO_TRUE@' ""

      # Set the PATH properly.
      substituteInPlace pm/pm-functions.in --replace '/sbin:/usr/sbin:/bin:/usr/bin' '$PATH:${binPath}:${sbinPath}'

      substituteInPlace src/pm-action.in --replace 'tr ' '${coreutils}/bin/tr '
      
      substituteInPlace pm/sleep.d/00logging --replace /bin/uname "$(type -P uname)"
    '';

  meta = {
    homepage = http://pm-utils.freedesktop.org/wiki/;
    description = "A small collection of scripts that handle suspend and resume on behalf of HAL";
    license = "GPLv2";
  };
}
{'fraction_non_alphanumeric': 0.1396598030438675, 'fraction_numerical': 0.02685765443151298, 'mean_word_length': 3.138888888888889, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 2}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '3737856', 'n_tokens_mistral': 842, 'n_tokens_neox': 779, 'n_words': 177}